Základní funkce a příkazy

Maple jako kalkulátor

1. Červeně je zadání v jazyku systému MAPLE, černě jsou komentáře, modře pak výsledky
2. Příkaz necháme provést tak, že umístíme kursor (karetu, to co bliká) na červený text a stikneme ENTER. Provedou se všechny příkazy v jednom oddílu vyznačeném vlevo svorkou [
3. Více příkazů vložíme navzájem odělěných střednímem (;) příp. dvojtečkou (:), pro přehlednost
   můžeme psát na více řádků pomocí <SHIFT-ENTER>.
4. Prázdný oddíl vložíme pomocí <CTRL-J> a <CTRL-K> případně myší a čudlíkem [>] nahoře.

```
>
```

```
> H:=sin(x)/cos(x);
```
Příkaz je nezbytné ukončit středníkem

$$H := \frac{\sin(x)}{\cos(x)}$$

```
> diff(H,x);
```
Při derivování se provedou jen nezbytné operace

$$1 + \frac{\sin(x)^2}{\cos(x)^2}$$

```
> simplify(%);
```
Očekávaný výsledek dostneme až pozjednodušení posledního výsledku (%)

$$\frac{1}{\cos(x)^2}$$

```
> H1:=int(%%,x);
```
Neurčitý integrál z předposledního výsledku (%%)

$$H1 := \tan(x)$$

```
> H1-H;
```

$$\tan(x) - \frac{\sin(x)}{\cos(x)}$$

Nula vyjde až po zjednodušení, které užívá pravidla pro práci s trigonometrickými funkcemi

```
simplify(H1-H);
```

$$0$$

Normal provádí zjednodušování jen s použitím pravidel pro práci s racionálními funkcemi

```
normal(H1-H);
```

$$-\frac{-\tan(x)\cos(x) + \sin(x)}{\cos(x)}$$

Ne vždy je výsledek **simplify** ten nejsrozumitelnější, někdy trochu pomůže funkce **factor**, která vytkne z výrazu co jde

Pozn. ***x*$4** je zkrácený zápis pro ***x,x,x,x*** -- počítáme čtvrtou derivaci

```
t1:=diff( (x^2-1)^8,x$4);
simplify(t1);
factor(t1);
```

$$t1 := 26880\,(x^2-1)^4\,x^4 + 16128\,(x^2-1)^5\,x^2 + 672\,(x^2-1)^6$$

$$43680\,x^{12} - 192192\,x^{10} + 332640\,x^8 - 282240\,x^6 + 117600\,x^4 - 20160\,x^2 + 672$$

$$672\,(65\,x^4 - 26\,x^2 + 1)\,(x-1)^4\,(x+1)^4$$

Použitím velkého počátečního písmene zabráníme provedení operací `Sum, Int, Diff, Product`

,
znak = je použit ve významu relace.

```
Sum(1/n^2,n=1..infinity)=sum(1/n^2,n=1..infinity);
```

$$\sum_{n=1}^{\infty} \frac{1}{n^2} = \frac{1}{6}\pi^2$$

Funkce **value** převede inertní formu **Sum** na **sum;** (**lhs** a **rhs** umožňují odkaz na levou resp. pravou starnu rovnosti)
pokud bychom místo čárky použili středník, % ve druhém výrazu budou odkazovat na spojení **Pravá_strana**

```
Pravá_strana=rhs(%),Levá_strana=value(lhs(%));
```

$$Pravá\_strana = \frac{1}{6}\pi^2,\ Levá\_strana = \frac{1}{6}\pi^2$$

Příklad integrálu, který MAPLE vypočíst neumí. (Vlastně umí, ale musí se přepsat na dva integrály do π/2).
Integrovanou funkci si označím ***G*,** protože ji ještě budu potřebovat. Zobrazit ji ale nepotřebuji, takže použiji **:** místo **;**

```
G:=1/sqrt(1+k*sin(phi)^3): iiG:=int(G,phi=0..Pi);
```

Protože výsledek obsahuje integrační znaménko, MAPLE při integraci neuspěl

$$iiG := \int_0^{\pi} \frac{1}{\sqrt{1 + k\sin(\phi)^3}}\, d\phi$$

V závislosti na povaze problému mohu nalézt cestu jak pokročit dál.
Nechť např. mohu rozvinout ***G***do řady v ***k*** (tedy předpokládám, že $|k|$<<1*)*

```
G1:=series(G,k,5);
```
Rozviň do 5-tého členu v k okolo nuly

$$G1 := 1 - \frac{1}{2}\sin(\phi)^3 k + \frac{3}{8}\sin(\phi)^6 k^2 - \frac{5}{16}\sin(\phi)^9 k^3 + \frac{35}{128}\sin(\phi)^{12} k^4 + O(k^5)$$

Nyní již integrace neselže

```
iG:=int(G1,phi=0..Pi);
```

$$iG := \pi - \frac{2}{3}k + \frac{15}{128}\pi k^2 - \frac{16}{63}k^3 + \frac{8085}{131072}\pi k^4 + O(k^5)$$

Protože výsledek je řada se zbytkem, neumí jej **plot** nakreslit a musím použít **convert**

```
plot(convert(iG,polynom),k=0..1/2);
```



Přestože ***iiG*** má formu integrálu, lze s ním pracovat podobně jako s jinými výrazy:

```
Diff(iiG,k)=diff(iiG,k);
evalf(eval(iiG,k=1/3));
```
evalf(F) a **evalf(F,digits)**

$$\frac{\partial}{\partial k}\int_0^{\pi}\frac{1}{\sqrt{1+k\sin(\phi)^3}}\,d\phi=\int_0^{\pi}-\frac{1}{2}\,\frac{\sin(\phi)^3}{(1+k\sin(\phi)^3)^{(3/2)}}\,d\phi$$

$$2.952758364$$

Následující obrázek ilustruje, jaké přesnosti dosahujeme rozvojem pro k=0..1/2

```
plot(convert(iG,polynom)-eval(evalf('iiG')),k=0..1/2,numpoints=12,
adaptive=false);
```

Tento příkaz obsahuje několik fint:

1. **eval(evalf('iiG'))** je (asi) nejkratší zápis, který odloží numerické vyhodnocení ***iiG*** až na okamžik, kdy už je dána numerická hondota ***k***
2. **numpoints=12,adaptive=false** jsou parametry, které zakazují počítat numerickou hodnotu integrálu ve více jak 12 bodech (kvůli rychlosti výpočtu)

Cvičení: nalezněte alternativu k fintě č.1, změňte počet členů rozvoje na 30 atp.



Při použití MAPLE jako kalkulátoru oceníme i schopnost řešit rovnice a soustavy rovnic

```
solve(tan(x+a)+tan(x-a)=y,x);
```

$$\arctan\left(\frac{1}{2}\,\frac{-2\tan(a)^2-2+2\sqrt{\tan(a)^4+2\tan(a)^2+1+y^2\tan(a)^2}}{y\tan(a)^2}\right),$$
$$\arctan\left(\frac{1}{2}\,\frac{-2\tan(a)^2-2-2\sqrt{\tan(a)^4+2\tan(a)^2+1+y^2\tan(a)^2}}{y\tan(a)^2}\right)$$

Pro řešení soustav musíme zadat sadu rovnic a neznámých pomocí {}
Složenými závorkami vytvoříme množinu prvků.

```
solve({x^2+x*y+y^2 = 1,a*x^2+y^2/a=1},{x,y});
```

$$\{y=\mathrm{RootOf}((1+a^4-a^2)\_Z^4+(2a^2-2a^3+a-2)\_Z^2+a^2-2a+1)\,(-a^3-1+a^2$$
$$+\mathrm{RootOf}((1+a^4-a^2)\_Z^4+(2a^2-2a^3+a-2)\_Z^2+a^2-2a+1)^2$$
$$+\mathrm{RootOf}((1+a^4-a^2)\_Z^4+(2a^2-2a^3+a-2)\_Z^2+a^2-2a+1)^2\,a^4$$
$$-a^2\,\mathrm{RootOf}((1+a^4-a^2)\_Z^4+(2a^2-2a^3+a-2)\_Z^2+a^2-2a+1)^2)/(a-1),$$
$$x=\mathrm{RootOf}((1+a^4-a^2)\_Z^4+(2a^2-2a^3+a-2)\_Z^2+a^2-2a+1)\}$$

Všechna řešení soustavy rovnic jsou zde representována pomocí **RootOf**.

To je někdy výhodné, protože s jedním výsledkem, který však reprezentuje všechny, lze obvykle pracovat snáze.

RootOf se zbavíme užitím **allvalues**.

```
Pruseciky:=simplify({allvalues( % )}):Pruseciky;
```

$$\left\{\left\{x=-\frac{1}{2}\frac{\sqrt{2}\sqrt{(1+a^4-a^2)\left(-2a^2+2a^3-a+2-\sqrt{a(4a^2-7a+4)}\right)}}{1+a^4-a^2}, y=\frac{1}{4}\frac{\left(a+\sqrt{a(4a^2-7a+4)}\right)\sqrt{(1+a^4-a^2)\left(-2a^2+2a^3-a+2-\sqrt{a(4a^2-7a+4)}\right)}\sqrt{2}}{(a-1)(1+a^4-a^2)}\right\}, \left\{x=-\frac{1}{2}\frac{\sqrt{2}\sqrt{(1+a^4-a^2)\left(-2a^2+2a^3-a+2+\sqrt{a(4a^2-7a+4)}\right)}}{1+a^4-a^2}, y=\frac{1}{4}\frac{\left(a-\sqrt{a(4a^2-7a+4)}\right)\sqrt{(1+a^4-a^2)\left(-2a^2+2a^3-a+2+\sqrt{a(4a^2-7a+4)}\right)}\sqrt{2}}{(a-1)(1+a^4-a^2)}\right\}, \left\{y=-\frac{1}{4}\frac{\left(a+\sqrt{a(4a^2-7a+4)}\right)\sqrt{(1+a^4-a^2)\left(-2a^2+2a^3-a+2-\sqrt{a(4a^2-7a+4)}\right)}\sqrt{2}}{(a-1)(1+a^4-a^2)}, x=\frac{1}{2}\frac{\sqrt{2}\sqrt{(1+a^4-a^2)\left(-2a^2+2a^3-a+2-\sqrt{a(4a^2-7a+4)}\right)}}{1+a^4-a^2}\right\}, \left\{x=\frac{1}{2}\frac{\sqrt{2}\sqrt{(1+a^4-a^2)\left(-2a^2+2a^3-a+2+\sqrt{a(4a^2-7a+4)}\right)}}{1+a^4-a^2}, y=-\frac{1}{4}\frac{\left(a-\sqrt{a(4a^2-7a+4)}\right)\sqrt{(1+a^4-a^2)\left(-2a^2+2a^3-a+2+\sqrt{a(4a^2-7a+4)}\right)}\sqrt{2}}{(a-1)(1+a^4-a^2)}\right\}\right\}$$

Následující obrázek ukazuje, že jde opravdu o řešení dané soustavy rovnic
Bohužel tato ilustrace vyžaduje několik nepřehledných konstrukcí, z nichž za zmínku stojí použití **subs** při další práci s výsledky **solve**, ostatní souvisí s příkazy pro malování, které budeme probírat později.

```
> subs(Pruseciky[2],[x,y]);
```

$$\left[-\frac{1}{2}\frac{\sqrt{2}\sqrt{(1+a^4-a^2)\left(-2a^2+2a^3-a+2+\sqrt{a(4a^2-7a+4)}\right)}}{1+a^4-a^2}, \frac{1}{4}\frac{\left(a-\sqrt{a(4a^2-7a+4)}\right)\sqrt{(1+a^4-a^2)\left(-2a^2+2a^3-a+2+\sqrt{a(4a^2-7a+4)}\right)}\sqrt{2}}{(a-1)(1+a^4-a^2)}\right]$$

```
> with(plots):
```

```
aa:=(k)->6/k:
```
Vybrané parametry *a(k)*

```
cc:=(k)->COLOR(RGB,0,k/20,8/k):
```
Barvy pro každý paramter *a(k)*

```
display(
```
Obrázek se skládá z více kusů, které chci zobrazit naráz

```
contourplot(x^2+x*y+y^2,x=-2..2,y=-2..2,contours=[1],color=RED),
```

```
seq(op(
```
Pro každý parametr *a* vykreslím množinu řešení druhé rovnice a přidám kolečka na místa průsečíků

```
[contourplot(aa(k)*x^2+y^2/aa(k),x=-2..2,y=-2..2,contours=[1],co
lor=cc(k)),
PLOT(
POINTS(seq(evalf(subs(subs(a=aa(k),Pruseciky[i]),[x,y])),i=1..no
ps(Pruseciky))),
cc(k),
SYMBOL(CIRCLE))]),
k=8..20));
```




```
>
```